
Introduction 
 
A lot of interest was shown in Chris G5VZ’s elegant and minimalistic design for a 
touch keyer, seen working at the UK National Hamfest 2013. 
 
As well as providing working touch paddles and iambic keyer, the “open” design 
allows experimentation with the firmware and hardware; additional features 
can be added such as buttons for sending messages from memory e.g. CQ calls. 
 
This special online edition of Key Note brings together the two articles covering 
the design and construction, first published in the Autumn and Winter 2012 
editions of Key Note. 

SILENT RUNNING 
A keyer with no moving parts 
 
At the end of April there was a flurry of activity on the FISTS Yahoo! 
Group: options for working silently from bed without disturbing a sleep-
ing partner.  There were some entertaining responses, but several FISTS 
members mentioned touch keyers. 
 
There are some nice touch keyers around but at a (not insignificant) cost. 
 
As it happens, the newest line of PICAXE chips supports touch input and 
the tiny but powerful SMD (surface mount) PICAXE 28X2 is available as a 
complete microcontroller implementation on a 28-pin DIL plug in.  (See 
Figure 1)  It includes voltage regulator, pro-
gramming socket and many other features.  
The AXE201 is available from Tech Supplies 
(techsupplies.co.uk) at £9.99 plus £2.50 de-
livery plus VAT, making a single item £14.99 
on your doormat.  Delivery is usually next 
day.  Other distributors are around but I 
cannot vouch for any of them: Tech Supplies 
seem reliable and fair. 
 
Known as AXE201, this 0.6” pitch, 28-pin 
gem is available for less than £10 ex-VAT and under fifteen pounds with 
tax and delivery included.  And unlike the SMD chip, the AXE201 doesn’t 
blow away when you sneeze.  All you need to program the AXE201 is a 
serial lead to connect its integral programming socket to your computer 
(such as the AXE027 USB lead) and free programming software (such as 
the BAS805 Programming Editor for Windows).  Touch paddles made 
from a couple of scraps of copper-clad PCB material and a speaker are all 
that’s required to turn it into a basic iambic keyer.  A transistor switch 
can be added to key a rig. 
 
This article covers the design basics and the construction of an iambic, 
mode A keyer with sidetone.  A follow-up piece will cover programming 
for other options and creating a Morse tutor.  Since the 28X2 spends 
most of its time idle and the keyer firmware occupies only a fraction of 

Chris Pearson, G5VZ / WG5VZ, #12540 

Figure 1: AXE201 module 

http://groups.yahoo.com/group/fistscw/
http://groups.yahoo.com/group/fistscw/
http://techsupplies.co.uk/

the available program space, the 
tutor and other features – including 
stored message sequences and 
suchlike – are easy additions once 
the basic hardware is put together. 
The circuit – showing the AXE201 as 
a single 28-pin box – is shown in Fig-
ure 3 (next page).  The addition of 
the LED is an optional flourish but it 
does provide a positive indication 
that power is present and the proc-
essor is processing.  Note that the 
dit paddle and the dah paddle are 
connected to the processor inputs 
with just a single wire: they float 
electrically, responding to the touch 
of an operator’s finger or thumb. 
 
Other than the paddles, the key 
components are the processor as-
sembly, a circuit board and optional 
28-pin socket and a speaker for side
-tone.  I’ve used a switched 3.5mm 

mono jack socket as a way to con-
nect headphones and mute the 
speaker – for fully silent running, 
omit the speaker and use only 
phones.  A quarter-inch jack to con-

Figure 2: Main components 

TOUCH SENSORS 
Touch sen-
sors gener-
ally replace 
push 
switches and 
provide switching options without 
moving parts.  Touch sensors make 
ideal silent paddles. 
 

Using the PICAXE touch and 
touch16 commands on touch-
enabled chips, the sensor is usually 
a metal plate connected by a single 
wire to an analogue input pin.  Be-
cause the touch inputs are capaci-
tive, direct contact with the sensor 
is not required and the metal plate 
should, ideally, be covered with a 
thin, non-conductive material: 
plastic film or varnish. 
 

PICAXE chips 
All M2 PICAXE chips (08M2, 18M2) 
plus 28X2 and 40X2 support touch 
and the AXE201 28-pin package 
does, too.  Note that touch is sup-
ported by firmware release B.3 and 
the older AXE200 won't work in 
this application.  The AXE200 is 
available heavily discounted but 
has older B.2 firmware. 
 

Because the touch sensor is capaci-
tive, it is affected by circuit layout 
and by wiring around the circuit 

nect to the transmitter can also be added plus a DC power socket.  I 
haven’t included an on/off switch as the current drawn is tiny and it’s 
powered from the shack DC supply. 
 
Assembly 
Building the circuit is very straightforward.  Begin by securing the DIL 
socket and wiring pin 28 (Top right) to the power source of 9 to 13.8V 
DC.  Pin 27, right next door, goes to ground: 0V supply.  At that stage, 
connect the supply and check the voltage with a meter.  (The AXE201 is 
resilient to reverse voltage but let’s not get 
brave!) 
 
Wire the 330Ω resistor to pin 24 and its free 
end to the anode (Long wire) of the LED.  
Connect the LED short lead – its cathode – to 
ground.  This makes the LED active high. 
 
Next, connect the negative terminal of a 
10μF electrolytic capacitor to pin 23.  Its posi-
tive lead-out is connected to one terminal of 
a low-impedance speaker or to the head-

Figure 3: PICAXE AXE201 Keyer Circuit by Chris G5VZ, Version 1 

Figure 4: AXE201 fitted to board 

phone jack.  The second audio lead, from the speaker or the jack socket, 
is wired to ground. 
 
Finally we need to connect the dit paddle and the dah paddle to pins 13 
and 14, respectively.  
 
Paddles 
The paddles were cut using tin-snips from a small piece of single-sided, 
copper clad circuit board.  Fibreglass board will cut using snips; SRBP and 
other phenolic bases are best cut using a knife, saw or cutting wheel.  
Their size and shape can reflect your personal preferences but ensure 
they are cut out as mirror images. 
 
Double-sided foamy tape can be used to join them and provide separa-
tion.  I applied tape to the insulated side of one, aligned them both and 
pressed together.  Finishing with a file may be required.  Initially I bright-
ened up the paddles (Scratched the surfaces with a file, actually) which 
makes soldering connections easier.  As shown in the photos, I initially 
soldered wires to the paddles and used gravity, then Blu-Tack and finally 
elastic bands to hold the paddles down.  In the boxed version, the pad-
dles are soldered directly to the circuit board which, in turn, is bolted to 
the box. 
 
Once the paddles are assembled and connections are made, cover the 
finger-pieces with self-adhesive film or spray with varnish.  I used Just 
Crystal Clear aerosol varnish. 
 
Connect the paddles to the circuit board.   
Check the voltage – if you 
haven’t already done so – 
and check for shorts be-
tween any of the other 
connected pins.  Then 
remove power, plug the 
AXE201 assembly into the 
28-pin socket and re-
apply power.  +5V regu-
lated should be present 

Figure 5: Assembled board with paddles 

on pin 25 (Measured between pin 25 and pin 27, ground). 
 
Programming 
The program listings are shown but are also available to download from 
fists.co.uk/knadditional.html 
 
The AXE201 should now be connected to a PC using an AXE029 or 
equivalent serial cable and a short routine downloaded to the PICAXE to 
blink the LED and confirm operation. 
 
The LED is connected to pin 24 which, in turn, is routed to input/output 
B.7 of the PICAXE; the LED is connected through a current limiting resis-
tor to ground so the device is lit when the pin is taken high. 
 
The program is blink_LED.bas 

do 

 high b.7   ' Blink LED 

 pause 1000 

 low b.7  

 pause 50 

loop 
 
The LED will illuminate for about a second, blink off, then light for a fur-
ther second.  This tells us that the AXE201 is powered up and the proces-
sor is running. 
 
The paddle touch sense needs to be calibrated.  Another short program 
allows us to do that and this should be set for your project. 
 
Program set_touch.bas 
 
main: 

 touch16 A.1,w0 ; read value into w0 

 SERTXD ("dah w0 IS ", #w0, cr,lf) 

 touch16 A.0,w0 ; read value into w0 

 SERTXD ("dit w0 IS ", #w0, cr,lf) 

 pause 50 

 goto main 
 
This routine sends the values for each paddle to the PC terminal screen.  
Record the values with the paddles idle then touch each paddle and re-
cord the reading.  In the program below, use the difference (touch value 
– idle value) as the comparative constant, in this case, 600: 
 

http://fists.co.uk/knadditional.html
http://fists.co.uk/knadditional.html
http://fists.co.uk/knadditional.html

 touch16 A.0,w0    ; read value into w0 

 if w0 > 600 then   ; check for touched state 
 
Finally, download or copy the keyer program, keyer_01.bas.  This is the 
basic routine without features beyond iambic, mode A keying.  It is writ-
ten top-down so we can see exactly what happens.  The dit sounding and 
dah sounding are called as subroutines and here we would switch the TX 
line.  Like several other keyers, it sends FB on the sidetone when pow-
ered up or reset. 

#picaxe 28x2 

 

' * *** ********************************************************** 

' * *** PICAXE Morse Keyer 

' * *** Iambic A Version 1 

' * *** Chris Pearson, G5VZ 

' * *** May 2012 

' * *** ********************************************************** 

 

 symbol ditLength = 10 

 symbol dahLength = 30 

 symbol silencePeriod = 10 

 symbol betweenCharacters = 30 

 symbol sideTone = 100 

 symbol silentTone = 0 

 

' * *** ********************************************************** 

 

 high b.7  ' Blink LED 

 pause 100 

 low b.7 

 pause 50 

 high b.7 

' 

' Send FB 

' 

 sound B.6, (sideTone, ditLength) 

 sound B.6, (silentTone, silencePeriod) 

 sound B.6, (sideTone, ditLength) 

 sound B.6, (silentTone, silencePeriod) 

 sound B.6, (sideTone, dahLength) 

 sound B.6, (silentTone, silencePeriod) 

 sound B.6, (sideTone, ditLength) 

 sound B.6, (silentTone, betweenCharacters) 

 sound B.6, (sideTone, dahLength) 

 sound B.6, (silentTone, silencePeriod) 

 sound B.6, (sideTone, ditLength) 

 sound B.6, (silentTone, silencePeriod) 

 sound B.6, (sideTone, ditLength) 

http://fists.co.uk/knadditional.html

 sound B.6, (silentTone, silencePeriod) 

 sound B.6, (sideTone, ditLength) 

 

' * *** ********************************************************** 

 

 

main: 

 

do 

 touch16 A.0,w0 ; read value into w0 

 if w0 > 600 then 

  gosub senddit 

 else 

 

 endif 

  

 touch16 A.1,w0 ; read value into w0 

 if w0 > 600 then 

  gosub sendDah 

 else 

 

 endif 

 

loop 

 

end 

 

' * *** ********************************************************** 

 

' 

' Subroutines 

' 

 

senddit: 

 low b.7 

 sound B.6, (sideTone, dahLength) 

 high b.7 

 sound B.6, (silentTone, silencePeriod) 

 

 return 

 

sendDah: 

 low b.7 

 sound B.6, (sideTone, ditLength) 

 high b.7 

 sound B.6, (silentTone, silencePeriod) 

 

 return 

  

' * *** ********************************************************** 

If it all works then it’s time to 
put it all in a box.  Notice how 
the paddles have been soldered 
down to the circuit board giving 
a solid (And hopefully lasting) 
mount.  The circuit board has 
been bolted to the removable 
lid of the enclosure.  The 
speaker has been glued into the 
box using hot-melt adhesive 
and a speaker grille made from 

a small piece of Veroboard SRBP matrix. 
 
A slot has been cut in the box to allow 
the paddles to be accessed and a bezel 
gives the LED a bit of style.  (I’m easily 
impressed!)  I’ve included a fairly stan-
dard 2.1mm DC power socket on the 
rear of the enclosure and a 6.5mm (¼”) 
jack for the radio connection and – as 
previously mentioned – a 3.5mm switch 
jack socket for external speaker or 
phones if a sidetone is required. 
 
In the box I think it looks quite the part. 
 
The next instalment will cover speed 
control and adding the TX switching – 
there are plenty of spare pins on the 
PICAXE and lots of room on the PCB for 
a transistor and two 10k resistors.  We’ll 
also look at a Morse tutor function and, 
using the same technology, stored se-
quences for automated sending. 
 
Any questions or feedback: 
chris@g5vz.co.uk 

Figure 6: Fitting the keyer into a box 

Figure 7: Board secured to box lid 

Figure 8: Assembled in box 

mailto:chris@g5vz.co.uk

Ed: General information about PICAXE is available from www.picaxe.com. 
 
The PICAXE hardware and software is available from: Revolution Education Ltd., 
Unit 2 Bath Business Park, Foxcote Ave, Bath, BA2 8SF, United Kingdom. 
Web: www.techsupplies.co.uk E-mail: orders@techsupplies.co.uk 
Phone: +44 (0)1761 430044 
 
Prices given in the article and below include VAT but exclude delivery and are 
correct at the time of writing. 
 
For the touch keyer project, the following hardware can be used: 
 AXE201 PICAXE-28X2 module £11.99 tinyurl.com/axe201 
 AXE027 PICAXE USB Download Cable £11.99 tinyurl.com/axe027 
 
(They also supply components (e.g. capacitors, diodes and resistors) commonly 
used in PICAXE circuits at very reasonable prices.) 
 
Free programming software for Windows, Mac and Linux is available from: 
tinyurl.com/picaxesw 
 
If you are interested in experimenting with PICAXE, there are various kits, pack-
ages, PCBs and other items available—see tinyurl.com/picaxeparts. 

http://www.picaxe.com/
http://techsupplies.co.uk
mailto:orders@techsupplies.co.uk
http://tinyurl.com/axe201
http://tinyurl.com/axe027
http://tinyurl.com/picaxesw
http://tinyurl.com/picaxeparts

SILENT RUNNING – PART TWO 
Adding Rig Control and Keying Speed  
 
The story so far 
The first part of this article appeared in the 
Autumn 2012 Key Note and described the 
construction of a touch-activated iambic 
keyer.  The keyer uses PICAXE technology so 
its programming is relatively straightforward, 
using BASIC.  To further simplify the hard-
ware, the keyer is built around a PICAXE 
AXE201 module—a complete microcontroller implementation in a 28-pin 
DIL format. 
 
Where next? 
The next steps are to add the hardware necessary to key a transmitter 

and to provide a usable speed control.  
With those hardware additions made we 
can also tidy up the microcontroller code 
(page 30) and start using a text-to-Morse 
technique that will allow future develop-
ment of stored sequences (Memory 
messages) and use as a Morse tutor. 
As a final development, adding an exter-
nal speaker socket allows some addi-

tional flexibility in side-tone. 
 
Keying a transmitter 
There are several options that can be chosen to key a transmitter.  Ini-
tially a simple transistor switch was considered but Graham, G3ZOD sug-
gested that opto-isolation would provide a better solution.  Using an 
opto-isolator like the 4N25 separates the keyer electronics from the ra-
dio’s, avoiding earth loops and the potential for stray DC finding its way 
into expensive transceiver hardware. 
The schematic diagram on the next page shows the 4N25 opto-isolator 
connected to the rig keying pin, leg 22 or PICAXE B.5.  The 4N25 is a six-
pin package with pins 1 and 2 connected to an internal LED so the elec-

Chris Pearson, G5VZ/WG5VZ, #12540 

Chris G5VZ’s touch keyer 

Additional components: 
opto-isolator, resistor, pot and knob 

https://fists.co.uk/members/docs/keynotes/kn1012a5.pdf

tronics are exactly the same as the red LED we connected to B.7 in part 
one.  The LED in the 4N25 switches a transistor with its leads brought out 
on pins 4, 5 and 6.  (Pin 3 is unconnected.)  The electrical isolation be-
tween input and out is several thousand volts. 
 
The schematic on the right shows how a transistor might be used if a 
suitable opto-isolator isn’t available. 
 
The additional resistor and the 
4N25 DIL package can be 
mounted on the same matrix 
board as the AXE201 module 
and the leads from pins 4 and 5 
on the 4N25 taken to a jack 
socket mounted on the casing. 
This circuit should key most 
transmitters but please do 

Keyer schematic 

Alternative keying circuit using a transistor 

check that the transmitter you are keying can be switched by taking the 
control wire to ground.  Note that keying with either an opto-isolator or 
a transistor is unlikely to be compatible with older valve (tube) rigs that 
use, for example, cathode or grid block keying. 
 
The microcontroller activates the keying line by simply changing the state 
of output B.5 which is given the name txPin in the program code.  The 
key line is kept active while the side-tone is audible using the same tim-
ing parameters for both. 
 
Speed control 
Some microprocessor-controlled keyers use a menu 
system to set parameters and defining speed, which 
often seems much more complicated than its ana-
logue equivalent: a simple rotary pot.  Since the PI-
CAXE has the ability to read varying voltage levels 
and then convert them to digital form on its ADC 
inputs, we are able to use a traditional potentiome-
ter to set the keyer speed. 
 
Placing a 47k linear pot between the five-volt rail and ground allows only 
a very small current to flow while taking the slider of the potentiometer 
to ADC input lets us measure the position of the pot which is propor-
tional to the voltage on that input.  In the program listing the input is de-

fined as speedPot.  
When the voltage is 
at 5V the digital value 
is 255 and when it is 
at ground potential, 
0V, the value is zero. 
 
The value on ADC 
channel 2 is checked 
on every pass 
through the main 
program loop so the 
speed pot can be ad-
justed at any time. 

Pot connections for 
clockwise rotation 

increasing keyer speed 

Keying and speed control wired-up 

External speaker jack and side-tone 
The external speaker jack is wired in parallel with the internal speaker 
with the mute contact optionally disconnecting it.  Since the side-tone 
sound is used to time the rig keying, if no side-tone is required replace 
the sideTone variable assignment as follows: 

 
' * *** 
' * *** Control values 
' * ***  
 symbol sideTone = 100    ' Side tone note 
 symbol silentTone = 0    ' Silence 

with 
 

' * *** 
' * *** Control values 
' * ***  
 symbol sideTone = 0    ' Side tone silent 
 symbol silentTone = 0    ' Silence 

 
These developments make the PICAXE keyer a useful addition to the 
shack—a functional iambic key and keyer. 

Program listing keyer_axe201_05.bas 
 
This can be downloaded from fists.co.uk/knadditional.html 
 
#picaxe 28x2 
 
' * *** ******************************************************************************************* 
' * *** 
' * *** PICAXE Morse Keyer 
' * *** Iambic A Version 2.1 
' * *** Chris Pearson, G5VZ 
' * *** October 2012 
' * *** 
' * *** Version 2 updates: 
' * *** Rotary control for speed selection 
' * *** Output to key a transmitter 
' * *** Version 2.1: minor corrections - 
' * *** 1 LED Active high (Not low!) 
' * *** 2 Use keyRig flag so transmit control is selectable (eg FB is not transmitted!) 
' * *** 
' * *** ******************************************************************************************* 
' * *** 
' * *** Storage 
' * *** 
      EEPROM (%01001000, %00011000)             ' "FB" 
 
 
' * *** ******************************************************************************************* 
' * *** 
' * *** Inputs and outputs 
' * *** 
      symbol dahPin = 1                         ' Touch input - ADC channel 1 
      symbol ditPin = 0                         ' Touch input - ADC channel 0 
      symbol speedPot = 2                       ' Rotary speed setting: IO pin A.2 is ADC channel 2 
      symbol sidePin = B.6                      ' Side-tone audio output 
      symbol ledPin = B.7                       ' Panel LED - active high 
      low ledPin                                ' LED off 
      symbol txPin = B.5                        ' Default rig keyer output 
      low txPin                                 ' Rig control line inactive (Receive mode) 
 
' * *** 
' * *** Control values 
' * ***       
      symbol sideTone = 100                     ' Side tone note 
      symbol silentTone = 0                     ' Silence 
       

http://fists.co.uk/knadditional.html#kn1012

' * *** 
' * *** Variables 
' * ***       
      symbol storedByte = b0                    ' Used to read out of EEPROM store 
      symbol loopCount = b1                     ' Loop counter 
      symbol bitValue = b2                      ' Used in bit shifting 
      symbol characterCounter = b3              ' Used to count characters in a string 
      symbol keyRig = b4                        ' Flag determines whether to key attached radio 
      keyRig = 0                                ' Default setting is don't key transmit line 
      symbol startFlag = b5                     ' Flag start bit in character code 
      symbol touchValue = w10                   ' Touch sensor input value 
      symbol speedValue = b6                    ' Raw speed pot reading 
      symbol currentSpeed = b7                  ' Active raw speed value 
      currentSpeed = 0 
      symbol ditLength = b10 
      symbol dahLength = b11 
      symbol silencePeriod = b12 
      symbol betweenCharacters = b13 
 
 
' * *** ******************************************************************************************* 
' * *** 
' * *** Blink the LED 
' * *** 
      low ledPin                                ' LED off 
      pause 100 
      gosub blinkLED 
' * *** 
' * *** Send welcoming "FB" 
' * *** 
      gosub setSpeed                            ' Set Morse speed to current pot position 
      gosub sendFB 
 
' * *** ******************************************************************************************* 
' * *** 
' * ***      Main program loop 
' * *** 
 
main: 
 
      keyRig = 1                                ' All Morse now echoed to key rig output line 
 
do 
      touch16 dahPin,touchValue                 ' Read dah touch sensor into touchValue 
      ''sertxd("dahPin value is ", #touchValue, cr, lf) 
      if touchValue > 600 then 
            gosub sendDit 
      endif 
       
      touch16 ditPin,touchValue                 ' Read dit touch sensor into touchValue 
      ''sertxd("ditPin value is ", #touchValue, cr, lf) 
      if touchValue > 600 then 
            gosub sendDah 
      endif 
       
      gosub setSpeed                            ' Sending speed may be changed 
 
loop 
 
end 
 
' * *** 
' * ***      End of main program loop 
' * *** 
 
' * *** ******************************************************************************************* 
 
' * *** 
' * ***      Subroutines 
' * *** 
 
sendDit: 
' * ***      Subroutine to output a single dit 
      low ledPin 
      if keyRig = 1 then 
            high txPin                          ' Transmit line toggled when keyRig set high 
      end if 
      sound sidePin, (sideTone, ditLength) 
      high ledPin 
      low txPin 
      sound sidePin, (silentTone, silencePeriod) 
 
      return 
 
sendDah: 
' * ***      Subroutine to output a single dah 
      low ledPin 



Page 18    Key Note  Special Edition 28th September 2013 

 
      if keyRig = 1 then 
            high txPin                          ' Transmit line toggled when keyRig set high 
      end if 
      sound sidePin, (sideTone, dahLength) 
      high ledPin 
      low txPin 
      sound sidePin, (silentTone, silencePeriod) 
 
      return 
       
sendChar: 
' * ***      Subroutine to send one Morse character 
      startFlag = 0                             ' Start flag reset 
      for loopCount = 0 to 7 
            bitValue = storedByte % 2           ' Shift out LSB 
            if startFlag = 0 then 
                  if bitValue = 1 then 
                        startFlag = 1           ' Start of character data 
                  end if 
            else 
                  if bitValue = 1 then          ' Send element of character data 
                        ''sertxd("Dah",cr,lf) 
                        gosub sendDah 
                  else 
                        ''sertxd("Dit",cr,lf) 
                        gosub sendDit 
                  end if 
            end if 
            storedByte = storedByte / 2         ' Shift byte one bit right 
      next loopCount 
      ''sertxd("Character space", cr, lf) 
      gosub sendSpace                           ' Space between characters 
 
      return 
       
sendSpace: 
' * ***      Subroutine to wait inter-character space 
      sound sidePin, (silentTone, silencePeriod) 
       
      return 
       
sendFB: 
' * ***      Reads the welcome message (Default is "FB") from store and output the Morse 
      for characterCounter = 0 to 1             ' Two characters of "FB" 
            read characterCounter,storedByte    ' Read Morse character from store 
            ''sertxd("Byte read is ", #storedByte) 
            gosub SendChar                      ' and send it 
      next characterCounter 
 
      return 
       
blinkLED: 
' * ***      Briefly illuminate the LED 
      high ledPin                               ' LED on 
      pause 100 
      low ledPin                                ' LED off 
       
      return 
       
setSpeed: 
      readadc speedPot,speedValue               ' Speed pot processing 
      speedValue = 255 - speedValue             ' Reverse direction of rotary control (!) 
      if currentSpeed <> speedValue then       
            ''sertxd("Raw pot value changed to ", #speedValue, cr, lf) 
            currentSpeed = speedValue 
            ditLength = currentSpeed / 12       ' Scale the basic timing unit 
            if ditLength < 2 then 
                  ditLength = 2                 ' No extreme speeds! 
            end if 
            ''sertxd("Speed set to ", #ditLength, cr, lf) 
            dahLength = ditLength * 3 
            silencePeriod = ditLength 
            betweenCharacters = ditLength * 3 
      end if 
 
      return 
       
' * *** 
' * ***      End of program 
' * *** 
 
' * *** ******************************************************************************************* 


